

**PROGRAMME POUR PB-700**

# **DES LUTINS ENVAHISSENT L'ÉCRAN**

**CH**AQUE programme Basic tient du mouton à cinq pattes : non seulement il doit réaliser ce qu'on attend de lui, mais il doit également être le plus performant possible tout en occupant peu de place en mémoire... Ce dernier point est fondamental sur les ordinateurs de poche, qui ne sont pas toujours riches en Koctets. Sur tous les matériels, et sur le PB-700 en particulier, le fait de réaliser des programmes simples et concis présente de nombreux avantages : moins de temps passé au clavier, risques d'erreurs réduits, maintenance et modifications plus faciles, raisonnement plus pratique à suivre ou à expliquer...

■ Une fois lancé, le programme présenté ici propose un jeu : des « lutins » d'un type assez particulier envahissent l'écran par la droite. Il s'agit de les faire disparaître grâce à un petit rayon qui part de la gauche de l'écran et qui est dirigé à partir du clavier : P lance ce rayon, Q le déplace vers le haut et Z le déplace vers le bas. Si on le fait descendre alors qu'il est déjà en bas de l'écran, il réapparaît en haut. Et inversement. Chaque lutin touché donne un

Au départ, trois niveaux de difficulté sont proposés (de 1 à 3). La partie se ter-

mine dès qu'une rangée de lutins rejoint la colonne de gauche d'où part le rayon.

Ce programme pourrait n'être considéré que pour le résultat (passionnant) qu'il propose. Mais on peut aussi l'analyser afin de comprendre sa structure et surtout, de l'adapter à d'autres Basic que celui du PB-700.

Le programme se divise en cinq parties : initialisation (lignes 1 à 16), progression des lutins (lignes 18 à 25), déplacement du rayon (lignes 28 à 45), tir (lignes 48 à 70) et fin de partie (lignes 78 à 95).

Les tableaux de variables sont dimensionnés par la ligne 5 ; les lignes 10 à 15 servent à créer trois variables graphiques qui sont respectivement : une suite de symboles CHR\$(250) représentant les « lutins », une suite de tirets représentant le rayon qui va servir à faire disparaître les lutins, et une suite d'espaces qui permettra d'effacer ce rayon. Le programme taillera dans ces variables des longueurs correspondant à ses besoins avec une instruction MID\$(lignes 20, 55 et 60).

## **Ligne remplie, partie finie**


Les lutins sont figurés par le symbole CHR\$(250) : une croix dans un carré ouvert (il manque un côté). Selon le niveau (T) choisi ligne 16, ils apparaissent par paquets de 1, 2 ou 3.

N est une variable aléatoire qui peut prendre les valeurs 0, 1, 2 ou 3. On peut ainsi déterminer sur quelle rangée de l'écran vont apparaître les lutins, à chaque passage du pointeur ligne 20. Par exemple, en niveau 2, si N = 3, nous verrons arriver deux symboles CHR\$(250) sur la rangée du bas de l'écran.

La ligne 25 donne le signal de la fin d'une partie : il y a plus de 17 lutins sur une ligne.

Une petite accolade, CHR\$(125), figure le départ du rayon ; elle se déplace verticalement sur la première





colonne de l'écran : ce qui revient à dire que ses coordonnées sont toujours sous la forme (0, X) avec X variant de 0 à 3. Quand la touche Z est activée (ligne 31), l'accolade se déplace vers le haut car X est incrémenté d'une unité ; inversement, une pression sur Q (ligne 30) la fera descendre d'un cran.

### Un tour d'écran

La ligne 35 mérite d'être soulignée. Grâce à elle, en effet, l'accolade peut « faire le tour » de l'écran : si on la fait descendre alors qu'elle est en bas de l'écran, elle réapparaît en haut, et vice-versa. Cette ligne 35 est en fait un double test que l'on aurait pu écrire comme ceci : 35 IF X = -1 THEN X = 3 (passage du haut en bas de l'écran) et 36 IF X = 4 THEN X = 0 (passage du bas en haut de l'écran).

Quand X est ainsi déterminé, la ligne 40 affiche l'accolade à l'endroit voulu, sa position précédente ayant été effacée par la ligne 30.

S'il n'y a pas de tir à ce moment (ligne 45), les lutins continuent de se promener impunément (ligne 20 et suivantes).

Quand la touche P est activée, les lignes 50 à 60 interviennent pour déclencher le rayon. Ligne 50, on incrémente le score S en passant, et l'on s'assure que le rayon ne risque pas de déborder de l'écran : c'est ce qui se passerait si l'on tirait sur une rangée entièrement vide (A(X) = 0).

La ligne 55 forme le tir : le rayon est constitué d'une portion de la variable B\$(1) (suite de signes "-") ; la longueur nécessaire est calculée grâce à A(X), de telle sorte que le rayon s'arrête sur le premier lutin de la rangée considérée.

La ligne 60 efface ce tir et le lutin touché : même principe d'action que la ligne 55, mais en remplaçant les signes "-" par des espaces. Puis l'on repart ligne 20 pour une nouvelle vague de lutins.

Lorsque les lutins ont réussi à atteindre l'accolade (signal donné par la ligne 25), ils claironnent leur victoire par des BEEP retentissants (lignes 80 et 85). Il ne reste plus qu'à afficher le score S (ligne 90) et, éventuellement, à repartir pour de nouvelles aventures (ligne 95) en appuyant sur une touche.

André REEB-GRUBER

```

1 REM *---INITIALISATION ---*
5 CLEAR :DIM A(4),A$(1)*17,B$(1)*17,
  C$(1)*18:CLS
10 FOR I=1 TO 17
11 A$(I)=A$(I)+CHR$(250):NEXT I
13 B$(1)="-----"
15 C$(1)=" "
16 INPUT "NIVEAU(1,2,3)";T:CLS
18 REM *---PROGRESSION DES LUTINS---*
20 N=INT(RND*4):A(N)=A(N)+T:LOCATE 19
  -A(N),N:PRINT MID$(A$(1),1,A(N)):
25 IF A(N)>17 THEN 80
28 REM *---DEPLACEMENT DU RAYON ---*
30 LOCATE 0,X:PRINT " ";IF INKEY$="Q
  " THEN X=X-1
31 IF INKEY$="Z" THEN X=X+1
35 IF X=-1 THEN X=3:IF X=4 THEN X=0
40 LOCATE 0,X:PRINT CHR$(125):
45 IF INKEY$<>"P" THEN 20
48 REM *--- TIR ---*
50 S=S+1:IF A(X)=0 THEN A(X)=1
55 A(X)=A(X)-1:LOCATE 1,X:PRINT MID$(
  B$(1),1,18-A(X)):
60 LOCATE 1,X:PRINT MID$(C$(1),1,18-
  A(X)):BEEP
70 GOTO 20
78 REM *--- FIN DE PARTIE ---*
80 FOR I=1 TO 10
85 BEEP 1:BEEP :BEEP 1:NEXT I
90 CLS :LOCATE 5,2:PRINT "SCORE:";S:
95 IF INKEY$=" " THEN 95 ELSE 5

```

**Lutins en nombre**  
Programme pour PB-700  
Auteur André Reeb-Gruber  
Copyright LIST et l'auteur

#### Liste des variables

A(N)	longueur d'une rangée de lutins
A\$(1)	suite de 17 symboles « lutins »
B\$(1)	suite de 17 signes "-" représentant le rayon
C\$(1)	suite de 18 espaces (effacement du rayon)
N	variable aléatoire de 0 à 3
T	niveau de difficulté (progression des lutins)
X	ordonnée du rayon

Les lutins arrivent à droite de l'écran ; le rayon qui part de la gauche essaie de contenir leur poussée

